By trial and error, the optimum diameter $D^*$ is found and the corresponding values of $L^*$ and $t^*$ are obtained by substitution. Thus

$$D^* = 1.277 \text{ m}$$

$$L^* = 10.860 \text{ m}$$

$$t^* = 12.77 \text{ mm}$$

$$T^* = 40 \text{ mm.}$$

The total cost for the optimal design is made up as follows:

| | |
|---|---|
| Cost of steel (cylinder) | \$16 691.56 |
| Cost of steel (hemisphere) | \$3925.31 |
| Cost of welding | \$1321.88 |
| Cost of insulation | \$4284.97 |
| Discounted losses | \$3225.56 |
| Total | \$29 449.28. |

This design is not practically feasible, since steel plate is available only in discrete thicknesses. If the value of plate thickness $t$ is rounded up to 13 mm, some advantage may be taken of the increased hoop tension strength to increase the diameter and thus reduce the surface area of the vessel.

From (12.13), if $t = 13$ mm, the maximum diameter is 1.3 m. Then by the volume constraint of (12.15), the length $L$ is 10.434 m. The modified design is then

$$D = 1.3 \text{ m}$$

$$L = 10.434 \text{ m}$$

$$t = 13 \text{ mm}$$

$$T = 40 \text{ mm}$$

and the total cost is increased to \$29 454.11. This last calculation suggests an alternative approach to the problem. Since $T$ is obtained by sub-optimization and the two equality constraints are relatively simple, it would be possible to develop an algorithm in which the plate thickness $t$ (mm) is the only independent variable. From $t$, the value of $D$ is found by (12.13) and, hence, $L$ from (12.15). All quantities and costs are then calculable.

## 12.5 A NEW WATER SUPPLY

### 12.5.1 Background

The problem described in this section is somewhat similar to the Thirstville 'case-study' of Section 1.4. The supply is assumed to be by gravity main, but the problem is complicated by the irregular demand pattern and a more detailed design of the balancing tank. The scene is set by the following memorandum, together with the typical cross-section of Figure 12.10.



To: Dr S. T. Mater, Civil Engineering New Works.
From: Ms Chris Talgazing, Planning Department.
Re: *New Water Supply.*

I have now received from the Process Planning Section an estimate of the water supply which will be required by the new plant. The demand, averaged over four-hour intervals, is given for a one-week cycle in the attached table, from which you will note that considerable fluctuation in demand is to be expected.

The local authority has assured me that the town supply main can provide in excess of 0.1 $m^3/s$ and I understand that the main passes within 3.2 kilometres of the demand point. The pressure elevation at the take-off point on the town main should be about 30 m above ground elevation at the plant.

I should like you to examine the costs of providing a water supply, including, if necessary, a reinforced concrete balancing tank. A typical cross-section of a similar reservoir is shown in the accompanying sketch; the same criteria for earth cover, ground slope, freeboard, etc., should be used in your estimate. I should point out, however, that the only ground available for such a tank is a long, level strip only 25 m wide.

I realize that there may be further information necessary before your study can be completed, and that estimates of construction costs are approximate. However, I hope your report will help to identify and define the main factors to be considered and show whether or not a balancing tank is justified.

*Expected Water Demand*
(flows in cubic metres per second, averaged over four-hour periods)

| Time | 12–4 a.m. | 4–8 a.m. | 8–12 noon | 12–4 p.m. | 4–8 p.m. | 8–12 midnight |
|---|---|---|---|---|---|---|
| Sun. | 0.012 | 0.020 | 0.037 | 0.031 | 0.020 | 0.017 |
| Mon. | 0.012 | 0.034 | 0.083 | 0.068 | 0.057 | 0.034 |
| Tues. | 0.023 | 0.040 | 0.068 | 0.062 | 0.045 | 0.028 |
| Wed. | 0.021 | 0.045 | 0.051 | 0.034 | 0.034 | 0.016 |
| Thurs. | 0.023 | 0.014 | 0.034 | 0.048 | 0.054 | 0.040 |
| Fri. | 0.021 | 0.014 | 0.034 | 0.048 | 0.054 | 0.040 |
| Sat. | 0.018 | 0.026 | 0.040 | 0.060 | 0.045 | 0.026 |

### 12.5.2 Problem formulation

The first step is to prepare a sketch of the system showing the relevant components, system parameters, and design variables. Figure 12.11 shows such a diagram, and the following system parameters and design variables are identified:




Figure 12.10 Typical section of in-ground tank




Figure 12.11 Diagrammatic sketch of the system (New Water Supply Problem Section 12.5)

System parameters:

| | | |
|---|---|---|
| Pipeline length | $L$ (m) | 3200 m |
| Available pressure head | $h_f$ (m) | 30 m |
| Available ground width | $W$(m) | 25 m |
| Outflow (demand) | $Q_{out}$ (m$^3$/s) | (see table) |
| Embankment slope | $M$ | 2:1 |
| Reinforced concrete bending modulus | $\frac{RM}{bd^2}$ (N/mm)$^2$ | 0.4 N/mm$^2$ |

Design variables:

| | |
|---|---|
| Inflow (supply) | $Q_{in}$ (m$^3$/s) |
| Pipe diameter | $D$ (m) |
| Inside tank breadth | $B$ (m) |
| Inside tank length | XL (m) |
| Water depth | $H$ (m) |
| Depth in ground | $G$ (m) |
| Concrete wall thickness | $d$ (m) |
| Tank volume | $V$(m$^3$) |

The objective function which is to be minimized comprises only capital expenditures. No continuing costs are included.

Objective function $z =$ $C1$ (Excavation) (12.18)
$+ C2$ (Embankment)
$+ C3$ (Import fill/export surplus)
$+ C4$ (Reinforced concrete)
$+ C5$ (Formwork — outside, inside, and roof slab)
$+ C6$ (Pipeline).



The following costs are assumed for the purpose of the analysis.

| | | |
|---|---|---|
| Excavation | \$5.00/m$^3$ | (12.19) |
| Form embankment | \$2.00/m$^3$ | |
| Import fill or dispose of surplus | \$2.50/m$^3$ | |
| Outside formwork | \$12.00/m$^2$ | |
| Inside vertical formwork | \$18.00/m$^2$ | |
| Suspended slab formwork | \$25.00/m$^2$ | |
| Reinforced concrete | \$100.00/m$^3$ | |
| Pipeline | $C6 = LD(390 - 11.5\sqrt{D})$. | |

### 12.5.3 Identifying constraints

The eight design variables defined in Section 12.5.2 are not independent and the next step is to determine the interactions which exist between these variables and to define the relevant constraints.

*Pipeline capacity*

The inflow $Q_{in}$ and pipe diameter $D$ must be related to the available piezometric gradient, which in turn is defined by the pressure head $h_f$ and the length $L$. A decision is needed as to the flow resistance law and the relevant friction loss parameters. The Strickler equation will be used here with an equivalent roughness height of $k = 0.3$ mm. (This is another system parameter, omitted from the list of Section 12.5.2.) Thus,

$$Q_{in} = \frac{8.41\sqrt{g}}{k^{1/6}} \frac{\pi}{4} D^2 \left(\frac{D}{4}\right)^{2/3} \left(\frac{h_f}{L}\right)^{1/2}. \tag{12.20}$$

*Balancing tank volume*

Within certain limits, the required storage volume $V$ will be dependent on the value of the inflow $Q_{in}$ and the specified demand pattern $Q_{out}$ (Figure 12.12).




Figure 12.12 Storage volume as a function of inflow

Inspection of the table of $Q_{out}$ values shows that if $Q_{in} \geqslant 0.083$ m$^3$/s, there will be no need for a balancing tank. This, however, might require a large and expensive pipeline. At the other extreme, the value of $Q_{in}$ must not be less than the average demand. This is given by

$$(Q_{out})_{ave} = \frac{1}{42}\sum_{i=1}^{42}(Q_{out})_i = 0.038 \text{ m}^3/\text{s}.$$

When $Q_{in} = 0.038$ m$^3$/s, the balancing storage required will be a maximum. For intermediate values of $Q_{in}$ (i.e. $0.038 \leqslant Q_{in} \leqslant 0.083$), some form of calculation or interpolation will be required.

*Tank dimensions*

The design variables include all three internal dimensions for the storage tank, as well as the required volume. Obviously, there is a simple relation between these quantities, i.e.

$$B \times XL \times H = V. \qquad (12.21)$$

*Wall thickness*

For this example, it will be assumed that the reinforced concrete wall of the tank will experience the greatest bending moment when the tank is hydraulically tested before backfilling on the outside and before completion of the roof slab. Figure 12.13 shows this condition. The thickness of the wall will be based on the




Figure 12.13 Hydrostatic loading of an unpropped cantilever

value of the bending moment on a simple, unsupported cantilever subject to hydrostatic loading. The analysis is more complex than this, but the assumption is probably adequate for the purpose of proportioning the tank.

For a fluid specific weight of $\gamma = 9810$ N/m$^3$ and depth $H$, the pressure at the base is $p = \gamma H$. The total force is given by

$$P = Hp/2 = \gamma H^2/2 \qquad (12.22)$$

Thus,

$$\text{BM} = PH/3 = \gamma H^3/6. \qquad (12.23)$$



The wall thickness can now be determined by the relation

$$\text{BM} = \text{RM} = Kbd^2 \qquad (12.24)$$

in which the flexural strength factor $K$ is given a low value of 0.4 N/mm$^2$ in order to reduce the risk of the concrete cracking on the wet, tension side. The quantity $b$ in (12.24) represents the breadth of the reinforced concrete section, but in this case the wall may be designed for a unit width of 1 m so that $b = 1$.

*Width constraint*

If a balancing tank is to be constructed, the total width between the toes of the embankment on each side must be less than 25 m. Clearly, this distance will depend on the tank dimensions $H$ and $B$, the wall thickness $d$, and the depth $G$ to which the tank is sunk in the ground. Figure 12.14 shows the geometry of the cross-section.




Figure 12.14 Relation between total width $W$ and other variables

The embankment height is given by

$$X = H + 1.43 - G \qquad (12.5)$$

in which the number 1.43 is the sum of the fixed quantities shown in Figure 12.10. The total width $W$ is then found as

$$W = B + 2d + 2mX \qquad (12.26)$$

and the necessary constraint takes the form

$$W - 25.0 \leqslant 0.0. \qquad (12.27)$$

### 12.5.4 Solving the mathematical model

From the preceding sections, the mathematical model may be set up.

$$\underset{Q^*, D^*, B^*, XL^*, H^*, G^*, d^*, V^*}{\text{Minimize } z} = C1 + C2 + \ldots + C6 \qquad \text{(refer 12.18)}$$

subject to

$$g_1(Q_{in}, D) = 0 \qquad \text{(refer 12.20)}$$

$$g_2(Q_{in}, Q_{out}, V) = 0$$

$$g_3(B, XL, H, V) = 0 \quad \text{(refer 12.21)}$$

$$g_4(d, H) = 0 \quad \text{(refer 12.22–24)}$$

$$(W - 25) \leqslant 0. \quad \text{(refer 12.25–27)}$$

This is a non-linear problem involving eight design variables, four equality constraints, and one inequality constraint. The problem can be greatly simplified if a sub-set of the design variables is chosen so as to allow the equality constraints to be substituted in the objective function, thus reducing the complexity of the model.

If the selected independent design variables are diameter $D$, tank breadth $B$, depth $H$, and in-ground depth $G$, the equality constraints can all be incorporated into the objective function as follows:

1. Calculate $Q_{in} = \phi(D)$ by (12.20).
2. Find required storage volume $V = \phi(Q_{in})$.
3. Obtain inside tank length $XL = \phi(V, B, H)$ (12.21).
4. Calculate wall thickness $d = \phi(H)$ (12.22–24).

The reduced model now takes the following form:

$$\underset{D^*, B^*, H^*, G^*}{\text{Minimize}}\ z = (C1 + C2 + \ldots + C6) \quad (12.28)$$

subject to

$$W - 25.0 \leqslant 0.$$

This model can be further reduced to an unconstrained model by incorporating a penalty term to ensure that the inequality constraint is satisfied, i.e.

$$\text{Minimize } z' = (C1 + C2 + \ldots + C6) + \text{FAC}(W - 25)\delta \quad (12.29)$$

where $\delta = 1$ if $W - 25 > 0$ and $\delta = 0$ if $W - 25 \leqslant 0$.

The multiplier FAC in (12.29) should be large enough to ensure that the penalty term is significant in comparison with the real objective function $z$. Equation (12.29) may now be optimized by a non-linear algorithm such as the Hooke and Jeeves pattern search (i.e. subroutine HJMIN).

### 12.5.5 Calculating balancing storage

As discussed in Section 12.5.3, a method is required to compute the necessary balancing storage as a function of the inflow $Q_{in}$. A suitable subroutine BALNCE is illustrated in Figure 12.15 which should be self-explanatory.

It would be possible to include a call of this routine within the cost routine used by HJMIN, but this would be rather inefficient. A better arrangement would be to include in the driving program a series of calculations which would determine corresponding values of inflow $Q_{in}$ and storage volume $V$, which could then be



```
      SUBROUTINE BALNCE(QOUT,NQOUT,QIN,VOLUME)
C ***************************************************************
C  THE ROUTINE OPERATES ON AN ARRAY OF REQUIRED OUTFLOWS
C  TO DETERMINE THE NECESSARY BALANCING STORAGE VOLUME WHICH
C  IS REQUIRED IF THE SPECIFIED INFLOW VOLUME IS SUPPLIED.
C  IF THE INFLOW IS LESS THAN THE AVERAGE DEMAND A VERY
C  LARGE VOLUME (E.G. 10.0E20) IS RETURNED.
C    QOUT   = ARRAY OF SIZE (NQOUT) CONTAINING THE REQUIRED
C             TIME SERIES OF OUTFLOWS.
C    NQOUT  = NO. OF OUTFLOWS.
C    QIN    = SPECIFIED AVAILABLE INFLOW.
C    VOLUME = COMPUTED BALANCING STORAGE VOLUME REQUIRED.
C  THE UNITS USED MUST BE CONSISTENT THROUGHOUT.  THUS
C  THE VOLUME IS DEFINED IN TERMS OF THE TIME INCREMENT
C  USED TO DEFINE THE OUTFLOW TIME SERIES.
C ***************************************************************
      DIMENSION QOUT(NQOUT)
      SUMQ=0.0
      VOL=0.0
      VMIN=0.0
      DO 10 I=1,NQOUT
      DV=QIN-QOUT(I)
      SUMQ=SUMQ + QOUT(I)
C  TEST IF TANK IS FULL AND OUTFLOW .LE. INFLOW
      IF(DV.GE.0.0.AND.VOL.GE.0.0) GOTO 10
      VOL=VOL + DV
      IF(VOL.GT.0.0) VOL=0.0
      IF(VOL.LT.VMIN) VMIN=VOL
   10 CONTINUE
      VOLUME=-VMIN
C  CHECK THAT AVERAGE DEMAND IS AVAILABLE AT LEAST.
      QAVE=SUMQ/FLOAT(NQOUT)
      IF(QIN.LT.QAVE) VOLUME=10.0E20
      RETURN
      END
```

Figure 12.15 FORTRAN subroutine BALNCE

transferred to the cost routine for interpolation in much the same fashion as illustrated in Figure 12.12. Two points are worth noting:

(1) If the inflow $Q_{in}$ is less than the average demand, the routine BALNCE automatically sets the required volume to an arbitrarily high value. This is equivalent to adding a penalty term if the constraint $Q_{in} \geqslant 0.038$ is violated.
(2) In calculating the cost of the balancing tank, a check should be made that the required volume is finite. If $Q_{in} > 0.083\ m^3/s$, then $V = 0.0$ and all the calculations associated with the tank can be skipped.

### 12.5.6 Typical solution

A typical solution using routine HJMIN is presented in this section. As described in Section 5.7.1, the method requires a main driving program and an objective function subroutine and must be executed in conjunction with the routine HJMIN as listed in Appendix A. The two subprograms will be discussed separately.

The dimension statements define the various arrays required. Two of these hold the values of the design variables and the corresponding incremental values to be used in the local search procedure of HJMIN. These appear in the calling statement. The other arrays are needed to store the outflow time history and the computed values of $Q_{in}$ and $V$ used to define the curve of Figure 12.12. Other design variables, system parameters, design quantities, and rates are transferred between the main program and the cost subroutine by means of labelled COMMON blocks. Those variables which constitute input to the routine COST are initialized either by a DATA statement, by simple assignments, by input from the keyboard, or by calculation. The inflow–storage function is defined by a set of 11 coordinate pairs which are evaluated in a DO-loop. Note that the minimum inflow is set slightly below the average demand to ensure that an arbitrarily high storage quantity is assigned, thus serving as a penalty term.

After the call of HJMIN, the optimal values of the design variables, together with other relevant information, are output. Some of the output is conditional on the balancing tank being of finite size. It is convenient to introduce a loop in the main program to allow alternative starting values to be defined. This helps to confirm the existence of a global minimum.

The objective function subroutine COST contains identical COMMON blocks and relevant dimension statements, as in the main program. It is convenient (and marginally more efficient), to re-assign the design variables as simple variables, rather than elements of an array. The first step is to calculate the pipeline cost and calculate by interpolation the storage volume required for the pipeline capacity. If no storage is required, the objective function calculation ends here. However, for finite storage volumes, the design of the tank makes up the bulk of the coding. The details of the design and the calculation of quantities should be fairly obvious from the coding and comment statements.

When the total real cost is calculated, penalty terms are added which correspond to the remaining constraints on the solution. The principal one is the available width of ground, but other (perhaps superfluous) non-negativity constraints have been added to keep the tank dimensions positive. It is easy to overlook the fact that the extrapolation step of the algorithm might produce a negative value of a variable which in turn generates a 'negative cost'.

The solution given by the program of Figure 12.16 is summarized in the output shown in Figure 12.17.

### 12.5.7 Allowance for discrete variables

The solution developed in the previous section may be impractical since the pipe diameter is assumed to be a continuous variable. A more realistic solution would be to remove the diameter from the array of design variables and introduce a loop in the main program to allow a series of discrete, commercially available pipe sizes to be defined. For this diameter, the inflow capacity and thus the storage volume would be fixed and transferred through COMMON block to the subroutine. The program of Figure 12.16 could be forced to operate in this way by



```
      PROGRAM TANKEX
      DIMENSION VAR(4),DVAR(4)
C  SEE ROUTINE COST FOR DEFINITION OF DESIGN VARIABLES
      DIMENSION QOUT(42),QAR(11),VOLAR(11)
C  THESE ARRAYS USED TO DEFINE DEMAND AND INFLOW/STORAGE
C  RELATIONSHIP
      COMMON /DESIGN/QAR,VOLAR,L,HF,K,M,XL,VOL,W,D,EMBHT,
     +              QIN
      COMMON /RATES/CEXC,CEMB,CDIF,CCONC,CFORMO,CFORMI,CFORMS
      REAL K,L,M
      EXTERNAL COST
C  DEFINE DEMAND TIME HISTORY
      DATA QOUT/0.012,0.020,0.037,0.031,0.020,0.017,
     +          0.012,0.034,0.083,0.068,0.057,0.034,
     +          0.023,0.040,0.068,0.062,0.045,0.028,
     +          0.021,0.045,0.051,0.034,0.031,0.016,
     +          0.023,0.040,0.062,0.071,0.051,0.034,
     +          0.021,0.014,0.034,0.048,0.054,0.040,
     +          0.018,0.026,0.040,0.060,0.045,0.026/
C  RESTART WITH NEW INITIAL VALUES
    1 CONTINUE
C  DEFINE INITIAL VALUES FOR DESIGN VARIABLES AND INCREMENTS
      DISPLAY "SUPPLY INITIAL VALUES FOR B, H, G, DIA"
      ACCEPT VAR(1),VAR(2),VAR(3),VAR(4)
      IF(VAR(1).LE.0.0) STOP
      DO 5 J=1,4
    5     DVAR(J)=VAR(J)/25.0
C  DEFINE SYSTEM PARAMETERS
      L=3200.0
      HF=30.0
      K=0.0003
      M=2.0
C  DEFINE COST RATES
      CEXC=5.00
      CEMB=2.00
      CDIF=2.50
      CCONC=100.0
      CFORMO=12.00
      CFORMI=18.00
      CFORMS=25.00
C  GET AVERAGE AND MAXIMUM DEMAND FLOWS.
      QAVE=0.0
      QMAX=0.0
      DO 10 J=1,42
        QAVE=QAVE + QOUT(J)
        IF(QOUT(J).GT.QMAX) QMAX=QOUT(J)
   10 CONTINUE
      QAVE=QAVE/42.0
C  COMPUTE POINTS ON INFLOW/STORAGE CURVE
C  SET MIN. FLOW JUST BELOW QAVE TO ENSURE PENALTY
      QMIN=QAVE-0.001
      DO 20 J=1,11
        QIN=QMIN + (QMAX-QMIN)*FLOAT(J-1)/10.0
        CALL BALNCE(QOUT,42,QIN,VOL)
        QAR(J)=QIN
        VOLAR(J)=VOL*4.0*3600.0
   20 CONTINUE
```

Figure 12.16 (a) Main FORTRAN program for problem of New Water Supply

```
C  DEFINE PARAMETERS FOR ROUTINE HJMIN
      DATA RHO,EPS,NW/0.5,0.01,0/
      NMAX=1000
C
      CALL HJMIN(VAR,DVAR,4,ANS,RHO,EPS,COST,NW,NMAX)
C
C  NOW OUTPUT RESULTS
      WRITE(6,100)NMAX
      WRITE(6,101)ANS
      WRITE(6,102)VAR(4)
        WRITE(6,110)QIN
  100 FORMAT(18H SOLUTION FOUND IN,I5,11H ITERATIONS)
  101 FORMAT(15H MINIMUM COST=$,10X,F10.2)
  102 FORMAT(18H PIPE DIAMETER(M)=,7X,F10.3)
  110 FORMAT(16H AVERAGE INFLOW=,9X,F10.3)
C  SKIP REMAINING OUTPUT IF NO TANK REQUIRED.
      IF(VOL.GT.0.0) GOTO 30
      WRITE(6,103)
  103 FORMAT(25H NO STORAGE TANK REQUIRED)
      GOTO 1
   30 CONTINUE
      WRITE(6,104)VAR(1),XL,VAR(2)
      WRITE(6,105)VOL
      WRITE(6,106)VAR(3)
      WRITE(6,107)D
      WRITE(6,108)W
      WRITE(6,109)EMBHT
  104 FORMAT(19H TANK DIMENSIONS(M),6X,3F10.3)
  105 FORMAT(19H TANK VOLUME(CUB.M),6X,F10.3)
  106 FORMAT(21H TANK DEPTH IN GROUND,4X,F10.3)
  107 FORMAT(19H TANK WALL THKSS(M),6X,F10.3)
  108 FORMAT(15H WIDTH USED (M),10X,F10.3)
  109 FORMAT(18H EMBANKMENT HT.(M),7X,F10.3)
      GOTO 1
      END
```

Figure 12.16 (a) — continued

```
      SUBROUTINE COST(X,CST)
C*****************************************************
C  THIS ROUTINE DETERMINES AN ARTIFICIAL OBJ. FUN.
C  FOR THE TANK + PIPELINE WITH A PENALTY TERM FOR
C  THE AVAILABLE WIDTH CONSTRAINT.
C    X(1)   = INSIDE TANK BREADTH (B)
C    X(2)   = INSIDE DEPTH OF WATER (H)
C    X(3)   = DEPTH OF TANK IN GROUND (G)
C    X(4)   = PIPE DIAMETER. (DIA)
C  USES COMMON BLOCKS /DESIGN/ AND /RATES/
C*****************************************************
      DIMENSION X(4)
      DIMENSION QAR(11),VOLAR(11)
      COMMON /DESIGN/QAR,VOLAR,L,HF,K,M,XL,VOL,W,D,EMBHT,
     +       QIN
      COMMON /RATES/CEXC,CEMB,CDIF,CCONC,CFORMO,CFORMI,CFORMS
      REAL K,L,M
C  REDEFINE DESIGN VARIABLES FOR CONVENIENCE
      B=X(1)
      H=X(2)
      G=X(3)
      DIA=X(4)
```

Figure 12.16 (b) Objective function subroutine



```
C  CHECK D>0 AND GET PIPE COST
      IF(DIA.LT.0.0) DIA=0.0
      CPIPE=L*DIA*(390.0 - 11.5*SQRT(DIA))
      CST=CPIPE
C  CALC QIN BY STRICKLER EQN.
      CONST=8.41*SQRT(9.81)/(K**0.1667)
      QIN=CONST*0.785*DIA*DIA*(DIA/4.0)**0.667*SQRT(HF/L)
C  FIND HOW MUCH BALANCING STORAGE NEEDED WITH THIS FLOW.
      CALL INTER1(VOLAR,QAR,11,QIN,VOL)
C  IF ZERO VOLUME IGNORE TANK CALCS
      IF(VOL.LE.0.0) RETURN
C  GET TANK LENGTH AND DESIGN WALL THICKNESS
      XL=VOL/(B*H)
      BM=9810.0*H*H*H/6.0
      D=SQRT(BM/400000.0)
C  GET OVERALL TANK DIMENSIONS
      HOA=H + 0.83
      BOA=B + 2.0*D
      XLOA=XL + 2.0*D
C  GET QUANTITIES OF EXCAVATION AND EMBANKMENT
      VOLEXC=BOA*XLOA*G
      EMBHT=HOA+0.6-G
      IF(EMBHT.LT.0.0) EMBHT=0.0
      A1=BOA*XLOA
      A2=(BOA+M*EMBHT)*(XLOA+M*EMBHT)
      A3=(BOA+2.0*M*EMBHT)*(XLOA+2.0*M*EMBHT)
      VOLEMB=(EMBHT/6.0)*(A1 + 4.0*A2 + A3)
      HAG=HOA - G
      IF(HAG.LT.0.0) HAG=0.0
      VOLEMB=VOLEMB - HAG*BOA*XLOA
C  FIND DIFFERENCE BETWEEN EXC AND EMB VOLUME.
C  IGNORE BULKING
      VOLDIF=ABS(VOLEXC-VOLEMB)
C  GET CONCRETE VOLUME
      CONC=2.0*(BOA+XL)*HOA*D + 0.53*BOA*XLOA
C  NOW GET OUTSIDE, INSIDE AND SLAB FORMWORK
      FORMO=2.0*(BOA+XLOA)*HOA
      FORMI=2.0*(B + XL)*(H + 0.3)
      FORMS=B*XL
C  NOW CALCULATE COSTS FOR TANK CONSTRUCTION
      C1=VOLEXC*CEXC
      C2=VOLEMB*CEMB
      C3=CONC*CCONC
      C4=FORMO*CFORMO + FORMI*CFORMI + FORMS*CFORMS
      C5=VOLDIF*CDIF
      CST=CPIPE+C1+C2+C3+C4+C5
C
C  CHECK WIDTH OF GROUND USED AND ADD PENALTY TERM
      W=BOA + 2.0*M*EMBHT
      PENW=1.0E07*(W-25.0)
      IF(PENW.LT.0.0) PENW=0.0
C  ADD PENALTY TERMS FOR NON-NEGATIVE VARIABLES
      PENG=1.0E6*(-G)
      PENH=1.0E6*(-H)
      PENB=1.0E6*(-B)
      IF(PENG.LT.0.0) PENG=0.0
      IF(PENH.LT.0.0) PENH=0.0
      IF(PENB.LT.0.0) PENB=0.0
      CST=CST + PENW + PENG + PENH + PENB
      RETURN
      END
```

Figure 12.16 (b) — continued

```
SUPPLY INITIAL VALUES FOR B, H, G, DIA ? 15.0, 2.5, 2.5, 0.2
SOLUTION FOUND IN 381 ITERATIONS
MINIMUM COST=$                 305617.00
PIPE DIAMETER(M)=                   .207
AVERAGE INFLOW=                     .046
TANK DIMENSION(M)                 15.010        23.278        2.860
TANK VOLUME(CUB.M)               999.126
TANK DEPTH IN GROUND               1.947
TANK WALL THKSS(M)                  .309
WIDTH USED (M)                    24.999
EMBANKMENT HT.(M)                  2.343
```

Figure 12.17 Results from program of Figure 12.16

setting DVAR(4) = 0.0 after the four quantities are initialized. Although computationally inefficient, this small change would cause the diameter to remain constant at the value input by the user. Typical results are shown in Figure 12.18.

```
SUPPLY INITIAL VALUES FOR B, H, G, DIA ?15.0, 2.0, 2.0, 0.2
SOLUTION FOUND IN  251 ITERATIONS
MINIMUM COST=$                 315963.19
PIPE DIAMETER(M)=                   .200
AVERAGE INFLOW=                     .042
TANK DIMENSIONS(M)                13.650         3.784        3.406
TANK VOLUME(CUB.M)              1477.795
TANK DEPTH IN GROUND               2.200
TANK WALL THKSS(M)                  .402
WIDTH USED (M)                    24.999
EMBANKMENT HT.(M)                  2.636
SUPPLY INITIAL VALUES FOR B, H, G, DIA ?15.0, 2.0, 2.0, 0.21
SOLUTION FOUND IN  207 ITERATIONS
MINIMUM COST=$                 305653.56
PIPE DIAMETER(M)=                   .210
AVERAGE INFLOW=                     .048
TANK DIMENSIONS(M)                14.555        20.091        3.179
TANK VOLUME(CUB.M)               929.732
TANK DEPTH IN GROUND               2.180
TANK WALL THKSS(M)                  .362
WIDTH USED (M)                    24.997
EMBANKMENT HT. (M)                 2.429
SUPPLY INITIAL VALUES FOR B, H, G, DIA ?15.0, 2.0, 2.0, 0.22
SOLUTION FOUND IN 203 ITERATIONS
MINIMUM COST=$                 306607.25
PIPE DIAMETER(M)=                   .220
AVERAGE INFLOW=                     .054
TANK DIMENSIONS(M)                14.775        14.774        3.009
TANK VOLUME(CUB.M)               656.897
TANK DEPTH IN GROUND               2.180
TANK WALL THKSS(M)                  .334
WIDTH USED (M)                    24.480
EMBANKMENT HT.(M)                  2.259
```

Figure 12.18 Results from program of Figure 12.16 modified for constant diameter



## 12.6 MINIMUM WEIGHT OF A PORTAL FRAME

### 12.6.1 Introduction

In Chapter 3 the minimum weight design of a rectangular portal frame was considered. In this section the problem is re-examined in a more general way with particular attention given to the following aspects of the problem:

(i) The dimensions of the frame should be variable and the type and intensity of the loading should be generalized.
(ii) The significance of the assumed linear relationship between the weight per unit length and fully developed plastic moment should be examined.
(iii) The effect on the optimal design of only discrete members being available should be studied.

The following memorandum provides background to the project:

*Memorandum*
To: Ben Tover, Drawing Office.
From: Willi Bendit, Fabricating Shop.
Re: *Minimum Weight Frames.*

We are anticipating enquiries regarding the supply of a number of rectangular portal frames. At the moment it is not clear what the dimensions will be, nor do we know the exact nature and intensity of the loading to be carried.

We have in stock a good selection of beam and column sections† and I should like to be in a position to respond quickly to any requests received.

I recall that on a previous occasion you developed a minimum weight design for a specific job subject only to concentrated loads, although I seem to remember that it was based on an assumption of linear relationship between section weight and plastic moment about which I had some doubts. Would you look into the possibility of preparing a computer program which would enable us to develop similar minimum weight designs for a variety of conditions?

### 12.6.2 Re-statement of the problem

For convenience the problem is re-stated as developed previously in Section 3.14. With reference to the frame of Figure 12.20(c) the printed problem may be written as follows for columns and beams of weight per unit length $W_c$ and $W_b$ respectively.

$$\text{Minimize } z = 2L_cW_c + L_bW_b \tag{12.30}$$

† Figure 12.19 shows the properties (mass per metre and plastic modulus) for the beam and column sections in stock.